@echo off
:: ============================================
:: IPAdapter模型下载工具（最终优化版）
:: ============================================
:: 设置UTF-8编码
chcp 65001 >nul
setlocal enabledelayedexpansion

:: 初始化工作目录
for %%i in ("%~dp0.") do set "WORKSPACE=%%~fi"
set "PYTHON=!WORKSPACE!\..\python\python.exe"

:: 路径1：固定路径
set "PATH1=!WORKSPACE!\..\Config\custom_nodes\ComfyUI_IPAdapter_plus"

:: 路径2：基础路径（用于动态匹配后缀）
set "BASE_PATH2=!WORKSPACE!\..\ComfyUI\custom_nodes\comfyui_ipadapter_plus"

:: 先检查固定路径1
if exist "!PATH1!\README.md" (
    set "FilePath=!PATH1!\README.md"
    goto FOUND_FILE
)

:: 动态检查路径2（支持后缀）
for /f "delims=" %%d in ('dir /ad /b "!BASE_PATH2!*" 2^>nul') do (
    if exist "!WORKSPACE!\..\ComfyUI\custom_nodes\%%d\README.md" (
        set "FilePath=!WORKSPACE!\..\ComfyUI\custom_nodes\%%d\README.md"
        goto FOUND_FILE
    )
)

:: 如果都没找到
echo [错误] 未找到Markdown文件，已检查以下路径：
echo 1. !PATH1!
echo 2. !BASE_PATH2!*
echo.
echo 请确认：
echo 1. IPAdapter是否安装在上诉路径中
echo 2. 确保存在README.md文件
pause
exit /b 1

:FOUND_FILE
echo.
echo [成功] 找到Markdown文件：
echo !FilePath!
echo.

:: 执行Python脚本
echo 正在启动下载流程...
@REM !PYTHON! -c "import sys; sys.path.append(r'!WORKSPACE!');"
"!PYTHON!" "!WORKSPACE!\ipadapter_down.py" "!FilePath!"

if errorlevel 1 (
    echo [错误] Python脚本执行失败 (错误码：%ERRORLEVEL%)
    pause
    exit /b 1
)

echo.
echo [操作成功完成]
pause